



第4章

单片机的定时/计数器与中断

4.1 定时/计数器概述

4.1.1 定时/计数器基本构成




图 4-1 51 单片机定时/计数器基本结构

4.1 定时/计数器概述

4.1.2 对T0和T1的控制方式

1. 控制寄存器TCON

表 4-1 TCON 的位结构

位地址	8FH	8EH	8DH	8CH	8BH	8AH	89H	99H
位名称	TF1	TR1	TF0	TR0	IE1	IT1	IE0	IT0

- 设置 TR0 (TR1) 为 0, 即停止 T0 (T1) 定时/计数器的工作。
- 设置 TR0 (TR1) 为 1, 即启动 T0 (T1) 定时/计数器开始工作。

4.1 定时/计数器概述

4.1.2 对T0和T1的控制方式

2. 工作方式寄存器TMOD




图 4-2 TMOD 的位结构

表 4-2 定时/计数器工作方式

M1 M0	工作方式	功能说明
0 0	方式 0	13 位定时/计数器
0 1	方式 1	16 位定时/计数器
1 0	方式 2	可自动重载的 8 位定时/计数器
1 1	方式 3	T0 分为两个 8 位定时/计数器。T1 无此工作方式

4.2 定时/计数器的4种工作方式

1. 工作方式0




图 4-3 定时/计数器工作方式 0

4.2 定时/计数器的4种工作方式

2. 工作方式1




图 4.4 定时/计数器工作方式 1

4.2 定时/计数器的4种工作方式

3. 工作方式2




图 4-5 定时/计数器工作方式 2

4.2 定时/计数器的4种工作方式

4. 工作方式3




图 4-6 定时/计数器工作方式3

4.3 定时/计数器T2

4.3.1 控制寄存器T2CON

表 4-3 T2CON 的格式

T2CON (C8H)	D7	D6	D5	D4	D3	D2	D1	D0
位名称	TF2	EXF2	RCLK	TCLK	EXEN2	TR2	C/ $\overline{T2}$	CP/ $\overline{RL2}$
位地址	CFH	CEH	CDH	CCH	CBH	CAH	C9H	C8H

1. 捕捉方式

2. 自动重载方式



4.4 定时/计数器使用示例

4.3.2 模式寄存器T2MOD

【程序 4-1】

```
                ORG      0000H
                MOV      TMOD, #00H    ; 此句可以不加
                MOV      TL1, #06H     ; 给TL1置初值立即数06H
                MOV      TH1, #0F8H    ; 给TH1置初值立即数F8H
                SETB     TR1           ; 启动T1工作
LR1:            JBC      TF1, LR2      ; 查询计数溢出否, 然后使TF1=0
                SJMP     LR1           ; 反复查询
LR2:            MOV      TL1, #06H     ; 重置初值
                MOV      TH1, #0F8H
                CPL      P1.0          ; 输出取反
                AJMP     LR1           ; 循环
```

4.4 定时/计数器使用示例

4.3.2 模式寄存器T2MOD

表 4-5 TMOD 的控制方式设置

TMOD	GATE	C/ \bar{T}	M1	M0	GATE	C/ \bar{T}	M1	M0
设置位	0	1	1	0	×	×	×	×

【程序 4-2】

```
                ORG      0000H
                MOV      TMOD, #60H    ; 设置T1为方式2
                MOV      TL1, #9CH     ; 给TL1置初值
                MOV      TH1, #9CH     ; 给TH1置初值
                SETB     TR1           ; 启动T1
LP1:            JBC      TF1, LP2      ; 查询计数溢出否
                SJMP     LP1
LP2:            CPL      P1.0         ; 输出取反
                AJMP    LP1           ; 循环
```

4.4 定时/计数器使用示例

4.3.2 模式寄存器T2MOD




图 4-7 外部正脉冲宽度测量

4.4 定时/计数器使用示例

4.3.2 模式寄存器T2MOD

【程序 4-3】

```
                ORG      0000H
                MOV      TMOD, #09H    ; 设置T0为方式1, GATE置1
                MOV      TL0, #00H     ; 将TL0清0
                MOV      TH1, #00H     ; 给TH0清0
WT1:            JB       P3.2, WT1     ; 等待INT0变低
                SETB    TR0           ; 启动T0
WT2:            JNB     P3.2, WT2     ; 等待正脉冲到
WT3:            JB       P3.2, WT3     ; 进入对机器周期计数阶段, 等待INT0变低
                CLR     TR0           ; 停止T0计数
                MOV     10H, TL0      ; 计数器T0低8位送入10H
                MOV     11H, TH0
RR:            SJMP    RR             ; 测量结束
```

4.5 单片机的中断系统

1. 中断技术实现分时处理多任务




图 4-8 中断响应示意图



4.5 单片机的中断系统

2. 中断技术可实现实时控制
3. 中断技术可对紧急事件进行优先处理



4.5 单片机的中断系统

4.5.1 单片机的中断源

(1) 外部中断类。

(2) 定时中断类。

(3) 串行中断类。

4.5 单片机的中断系统

4.5.2 中断控制寄存器的设置

1. 中断标志寄存器的设置

表 4-6 SCON 的格式

SCON	位地址							99H	98H
98H	位名称							TI	RI

2. 中断允许控制寄存器IE

表 4-7 IE 的格式

IE	位地址	AFH	AEH	ADH	ACH	ABH	AAH	A9H	A8H
A8H	位名称	EA	—	—	ES	ET1	EX1	ET0	EX0

4.5 单片机的中断系统

4.5.2 中断控制寄存器的设置

3. 中断优先级控制寄存器IP

表 4-8 寄存器的内容及位地址定义

IP	位地址	BFH	BEH	BDH	BCH	BBH	BAH	B9H	B8H
B8H	位名称	—	—	—	PS	PT1	PX1	PT0	PX0

表 4-9 同一级中断源优先顺序

中 断 源	同一级中断源优先顺序
外部中断源 INT0	高级 ↓ 低级
定时/计数器 T0 中断	
外部中断源 INT1	
定时/计数器 T1 中断	
串行通信口中断	
定时/计数器 T2 中断	

4.5 单片机的中断系统




图 4-9 51 单片机中断结构示意图



4.5 单片机的中断系统

4.5.3 中断响应过程

1. 中断采样
2. 中断查询
3. 中断响应

表 4-10 中断源与中断矢量地址表

中 断 源	中断矢量地址
外部中断 0 $\overline{INT0}$	0003H
定时/计数器 T0 中断	000BH
外部中断 1 $\overline{INT1}$	0013H
定时/计数器 T1 中断	001BH
串行通信口中断	0023H
定时/计数器 T2 中断	002BH



4.5 单片机的中断系统

4.5.3 中断响应过程

```
                ORG      0000H      ; 程序起始地址
                SJMP     MAIN        ; 进入主程序
                ORG      0003H      ; 定义响应外部中断0矢量地址
                LJMP     EXTINT      ; 转跳至外部中断0的服务程序
                ORG      0030H      ; 定义主程序起始地址
MAIN:           ...                ; 主程序开始
                ...
EXTINT:        ...                ; 外中断0之中断服务程序开始
                RETI              ; 中断返回
```

4. 中断响应处理



4.5 单片机的中断系统

4.5.4 中断请求的撤销方法

1. 外中断请求的撤销
2. 定时中断请求的撤销
3. 串行中断和T2中断请求的撤销

4.6 中断应用编程实例

【程序 4-4】

```
                ORG     0000H       ; 主程序开始
                SJMP    MAIN        ; 转主程序入口
                ORG     001BH       ; T1中断入口
                LJMP    T1_SP       ; 转T1中断服务程序
                ORG     0030H       ; 主程序入口
MAIN:           MOV     TMOD, #10H   ; 设T1为定时方式1
                MOV     TH1, #0FFH  ; 设TH1初值
                MOV     TL1, #06H   ; 设TL1初值
                SETB    TR1         ; 启动计数器T1
                SETB    ET1        ; 允许T1中断
                SETB    EA         ; 开放总中断口
                SJMP    $          ; 程序原地循环, 等待中断
T1_SP:         ; T1中断服务程序
                MOV     TH1, #0FFH  ; 重载TH1初值
                MOV     TL1, #06H   ; 重载TL1初值
                CPL     P1.0        ; 对P1.0求反, 输出方波脉冲
                RETI             ; 中断返回
                END
```


4.6 中断应用编程实例

【程序 4-5】

```
                ORG      0000H
                AJMP     MAIN          ; 转主程序入口
                ORG      001BH        ; T1中断入口
                LJMP     T1_SP        ; 转T1中断服务程序
                ORG      0030H        ; 主程序入口
KAIN:           MOV      TMOD, #20H   ; 设T1为定时方式2
                MOV      TH1, #06H   ; 设TH1初值
                MOV      TL1, #06H   ; 设TL1初值
                SETB     TR1          ; 启动计数器T1
                SETB     ET1          ; 允许T1中断
                SETB     EA          ; 开放总中断
                SJMP     $            ; 程序原地循环, 等待中断
T1_SP:          ; T1中断服务程序
                CPL      P1.0        ; 对P1.0求反, 输出方波脉冲
                RETI          ; 中断返回
                END
```

【程序 4-6】

```
ORG      0000H
SJMP    MAIN      ; 转主程序入口
ORG      000BH    ; T0中断入口
LJMP    T0_INT    ; 转T0中断服务程序
ORG      0030H    ; 主程序入口
MAIN:    MOV      TMOD, #01H    ; 设T0为定时器、工作方式1
         MOV      TH0, #3CH    ; 设TH0初值
         MOV      TL0, #0B0H   ; 设TL0初值
         MOV      30H, #0FEH   ; 置电灯起始循环常数
         SETB    TR0          ; 启动计数器T0
         SETB    ET0         ; 允许T0中断
         SETB    EA         ; 开放总中断
         SJMP    $           ; 程序原地循环。这里可以安排其他程序
T0_INT:  ; T0中断服务程序
         PUSH    ACC         ; 保护原来ACC中的数据
         MOV     TH0, #3CH   ; 重装定时器初值
         MOV     TL0, #0B0H
         MOV     A, 30H     ; 将电灯起始常数送入A
         RL      A          ; 每隔100ms循环左移一次
         MOV     P1, A      ; A中内容输出到P1口
         MOV     30H, A     ; 将驱动发光管循环常数放回30H中
         POP     ACC        ; 将保护好的数据返回ACC
         RETI             ; 中断返回
         END              ; 结束
```




图 4-10 发光二极管循环点亮



4.6 中断应用编程实例

【程序 4-7】

```
RTI:    PUSH    PSW          ; 压栈保护现场
        PUSH    A
        SETB   PSW.4        ; 寄存器工作区切换
        CLR    PSW.3
        CLR    TI
        MOV    A,    @R0
        MOV    C,    P          ; 加上奇偶校验位
        MOV    ACC.7, C
        MOV    SBUF,  A        ; 数据发送
        INC    R0
        POP    A              ; 恢复现场
        POP    PSW
        RETI                   ; 中断返回
```

4.6 中断应用编程实例





图 4-11 P1.0 输出 PWM 波形

【程序 4-8】

```
MOV     30H, #05      ; 设定低电平持续时间比例
MOV     31H, #95      ; 设定高电平持续时间比例
MOV     TMOD, #02H    ; T0工作方式2
MOV     TH0, #206     ; 设置定时器初值
MOV     TL0, #206
SETB    TR0           ; 启动定时器T0
L1:     MOV     R2, 30H ; 取高电平循环次数
        SETB    P1.0   ; P1.0输出高电平
L2:     JNB     TF0, L2 ; 查询定时器溢出标志
        CLR     TF0    ; 清除溢出标志
        DJNZ   R2, L2  ; 持续高电平输出循环
        MOV     R2, 31H ; 取低电平循环次数
        CLR     P1.0   ; P1.0输出低电平
L3:     JNB     TF0, L3 ; 查询定时器溢出标志
        CLR     TF0    ; 清除溢出标志
        DJNZ   R2, L3  ; 持续低电平输出循环
        SJMP   L1
```

【程序 4-9】



```
ORG      0000H
AJMP     START      ; 进入主程序
ORG      000BH      ; T0的中断入口
LJMP     TO_PROC    ; 转中断服务程序
ORG      0030H

START:
MOV      30H, #05    ; 设定高电平持续时间比例
MOV      31H, #95    ; 设定低电平持续时间比例
MOV      TMOD, #02H  ; T0工作方式2
MOV      TH0, #206   ; 设置定时器初值
MOV      TLO, #206
SETB     TRO         ; 启动定时器T0
SETB     ETO         ; 允许T0中断
SETB     EA         ; 开放总中断
SETB     00H        ; 设置高电平输出标志
SETB     P1.0       ; P1.0输出高电平
MOV      R2, 30H    ; 取高电平循环次数
NOP
SJMP     $

TO_PROC:
JNB     00H, PROC_L ; 若标志位为0, 转低电平处理
SETB     P1.0       ; P1.0输出高电平
DJNZ    R2, EXT     ; 持续高电平输出循环
MOV      R2, 31H    ; 取低电平循环次数
CLR      00H        ; 设置低电平标志

PROC_L:
CLR     P1.0        ; P1.0输出低电平
DJNZ    R2, EXT     ; 持续低电平输出循环
SETB     00H        ; 设置高电平标志
MOV      R2, 30H    ; 取高电平循环次数

EXT:    RETI        ; 中断返回
```